pragma solidity >=0.7.0 <0.8.0;
//SPDX-License-Identifier: MIT

// import "github.com/OpenZeppelin/openzeppelin-contracts/blob/release-v3.4-solc-0.7/contracts/access/Ownable.sol";
// import "github.com/OpenZeppelin/openzeppelin-contracts/blob/release-v3.4-solc-0.7/contracts/token/ERC20/IERC20.sol";
// import "github.com/OpenZeppelin/openzeppelin-contracts/blob/release-v3.4-solc-0.7/contracts/token/ERC20/SafeERC20.sol";

//OZ libraries:

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}


/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}



/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}


/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        uint256 c = a + b;
        if (c < a) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the substraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b > a) return (false, 0);
        return (true, a - b);
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) return (true, 0);
        uint256 c = a * b;
        if (c / a != b) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a / b);
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a % b);
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");
        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b <= a, "SafeMath: subtraction overflow");
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        if (a == 0) return 0;
        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");
        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: division by zero");
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: modulo by zero");
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        return a - b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryDiv}.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a % b;
    }
}


/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}


/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require((value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(value, "SafeERC20: decreased allowance below zero");
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) { // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// === end of libs ===

// NOTE for those re-using this contract code. If your token has implemented "emergency admin method" or similar,
//  that token can be withdrawn from this staking contract. Also please note that pausable tokens have impact
//  on operationability of staking.

contract Staking is Ownable {
    
    /*
     * Event fired when administrator withdraws tokens from reserve. 
     */
    event AdminWithdrawn(uint256 amount);
    
    /*
     * Event fired when indexes of stakes are re-organised for given user. Any caching mechanism that would cache
     *  methods that use stake id for this user must be invalidated.
     */
    event StakeReorg(address indexed user);
    
    /*
     * Signalizes start of staking for given user, selected plan ID and amount of staked tokens. 
     */
    event StakeCreated(address indexed user, uint256 indexed stakeId, uint256 stakeAmount);

    /*
     * Struct describing ongoing stake
     */
    struct Stake{
        uint256 unlockTime;
        uint256 stakeAmount;
        uint256 stakeReward;
        uint256 stakePlan;
    }

    /*
     * Struct describing stake plan.
     */
    struct StakePlan{
        // WEI of profit per TOKEN assumed 10 ** 18. This code is prepared to work for ID tokens,
        //  thus only tokens which have 18 decimals are supported.
        uint256 rewardPerToken;
        uint256 planTimeSeconds;
        bool enabled;
    }

    IERC20 private _token;
    uint256 private _lockedTokens;
    
    mapping (address => Stake[]) private _stakes;
    StakePlan[] private _plans;

    constructor(IERC20 token) {
        _token = token;
        _lockedTokens = 0;
    }
    
    /*
     * Checks if plan of index planId exists.
     */
    function planExists(uint256 planId) public view returns (bool) {
        return _plans.length > planId;
    }
    
    /*
     * Checks if user can currently create stake with this plan id.
     *  does not check availability of tokens for plan realisation - it's checked in stake method.
     */
    function planAvailable(uint256 planId) public view returns (bool) {
        return planExists(planId) && _plans[planId].enabled;
    }
    
    /*
     * Anwsers how much profit will be available after staking period is over. Does not include "base" stake amount
     */
    function getPlanProfit(uint256 planId, uint256 stakeSizeWei) public view returns (uint256) {
        require(planAvailable(planId), "Plan is unavailable");
        
        return (stakeSizeWei * _plans[planId].rewardPerToken) / 10 ** 18;
    }
    
    /*
     * Locks up user's funds in contract for period of time specified by plan.
     *  after period of time user can call withdrawAllAvailable method to withdraw all available for withdrawal stakes.
     */
    function stake(uint256 amount, uint256 planId) public {
        require(planAvailable(planId), "Plan is not available");
        
        uint256 profit = getPlanProfit(planId, amount);

        require(profit > 0, "Stake is too low to gain any profit. Refusing the stake.");
        require(_lockedTokens + profit <= _token.balanceOf(address(this)), "Guarantee reserve currently dry, cannot on-chain guarantee profit.");
        SafeERC20.safeTransferFrom(_token, msg.sender, address(this), amount);

        _lockedTokens += amount;
        _lockedTokens += profit;
        
        _stakes[msg.sender].push(
            Stake(
                block.timestamp + _plans[planId].planTimeSeconds,
                amount,
                profit,
                planId
            )
        );
        emit StakeCreated(msg.sender, planId, amount);
    }
    
    /*
     * Checks if provided address can currently withdraw any stake. If this method returns true user can call
     *  withdrawAllAvailable method.
     */
    function canWithdrawAny(address who) public view returns (bool) {
        for(uint256 i; i < _stakes[who].length; i++) {
            if(block.timestamp > _stakes[who][i].unlockTime) {
                return true;
            }
        }
        return false;
    }

    /*
     * Finds all stakes available for withdrawal for sender and transfers them with profits to sender.
     */
    function withdrawAllAvailable() public {
        require(canWithdrawAny(msg.sender), "No stakes can be withdrawn at this point...");
        
        bool stakeReorged = false;
        uint256 withdrawalAmount = 0;
        
        uint256 stakesLength = _stakes[msg.sender].length;
        
        for(uint256 i; i < stakesLength; i++) {
            if(block.timestamp > _stakes[msg.sender][i].unlockTime) {
                
                withdrawalAmount += _stakes[msg.sender][i].stakeAmount;
                withdrawalAmount += _stakes[msg.sender][i].stakeReward;
                
                
                //if last element of array
                if(i == stakesLength - 1) {
                    delete _stakes[msg.sender][i];
                } else {
                    _stakes[msg.sender][i] = _stakes[msg.sender][stakesLength - 1];
                    delete _stakes[msg.sender][stakesLength - 1];
                    stakeReorged = true;
                }
                
                _stakes[msg.sender].pop();
                stakesLength--;
            }
        }
        
        if(stakeReorged) {
            emit StakeReorg(msg.sender);
        }
        
        _lockedTokens -= withdrawalAmount;
        SafeERC20.safeTransfer(_token, msg.sender, withdrawalAmount);
    }
    
    /*
     * Returns amount of plans available. Used for iteration over plans to see what's available at the moment.
     */
    function getPlansCount() public view returns (uint256) {
        return _plans.length;
    }

    /*
     * Returns amount of total stakes of given address.
     */
    function getStakesCount(address who) public view returns (uint256) {
        return _stakes[who].length;
    }

    /*
     * Returns how much was staked for given address at given index.
     *  Important: do not cache. Indexes can change during withdrawal - in that scenario StakeReorg event is emited.
     *  StakeReorg effectively invalidates any cache mechanism for given address.
     */
    function getStakeAmount(address who, uint256 stakeIndex) public view returns (uint256) {
        require(_stakes[who].length > stakeIndex, "Invalid stake index");
        return _stakes[who][stakeIndex].stakeAmount;
    }
    
    /*
     * Returns unix timestamp when stake on given index is ready to be withdrawn.
     *  Important: do not cache. Indexes can change during withdrawal - in that scenario StakeReorg event is emited.
     *  StakeReorg effectively invalidates any cache mechanism for given address.
     */
    function getStakeUnlockTimestamp(address who, uint256 stakeIndex) public view returns (uint256) {
    
        require(_stakes[who].length > stakeIndex, "Invalid stake index");
        return _stakes[who][stakeIndex].unlockTime;
    }
    
    /*
     * Returns amount of reward that will be collected by given user on given index.
     *  Important: do not cache. Indexes can change during withdrawal - in that scenario StakeReorg event is emited.
     *  StakeReorg effectively invalidates any cache mechanism for given address.
     */
    function getStakeReward(address who, uint256 stakeIndex) public view returns (uint256) {
        require(_stakes[who].length > stakeIndex, "Invalid stake index");
        return _stakes[who][stakeIndex].stakeReward;
    }

    /*
     * Returns plan ID of stake used. Note that this plan may be not available anymore.
     *  Important: do not cache. Indexes can change during withdrawal - in that scenario StakeReorg event is emited.
     *  StakeReorg effectively invalidates any cache mechanism for given address.
     */
    function getStakePlan(address who, uint256 stakeIndex) public view returns (uint256) {
        require(_stakes[who].length > stakeIndex, "Invalid stake index");
        return _stakes[who][stakeIndex].stakePlan;
    }

    /*
     * Returns unix timestamp of when stake was registered.
     *  Important: do not cache. Indexes can change during withdrawal - in that scenario StakeReorg event is emited.
     *  StakeReorg effectively invalidates any cache mechanism for given address.
     */
    function getStakeStartedTimestamp(address who, uint256 stakeIndex) public view returns (uint256) {
        require(_stakes[who].length > stakeIndex, "Invalid stake index");
        return _stakes[who][stakeIndex].unlockTime - _plans[_stakes[who][stakeIndex].stakePlan].planTimeSeconds;
    }

    /*
     * Returns how much admin can safely withdraw. 
     */
    function getAdminWithdrawable() public view returns (uint256) {
        // if code reused for token other than ID, could be bad token that implements "emergency admin method" that will withdraw from any account
        require(_token.balanceOf(address(this)) >= _lockedTokens, "Critical bug in code or bad ERC token! Do not withdraw and resupply reserve ASAP!");
        return _token.balanceOf(address(this)) - _lockedTokens;
    }

    /*
     * Withdraws available tokens (cannot withdraw guarantee reserve for already ongoing stakes) from reserve
     *  Supplying 0 as amount withdraws all available tokens
     *  Owner only.
     */
    function adminWithdraw(uint256 amount) public onlyOwner {
        if(amount == 0) {
            amount = getAdminWithdrawable();
        } else {
            require(getAdminWithdrawable() >= amount, "Not enough tokens to withdraw");
        }
        require(amount > 0, "No tokens to withdraw");
        emit AdminWithdrawn(amount);
        SafeERC20.safeTransfer(_token, msg.sender, amount);
    }

    /*
     * Adds available plan for staking. Owner only.
     */
    function addPlan(uint256 rewardsForPlan, uint256 planTimeSeconds) public onlyOwner {
        require(rewardsForPlan > 0, "Plan will never be profitable");
        require(planTimeSeconds > 0, "Plan will finish instantly");
        _plans.push(StakePlan(rewardsForPlan, planTimeSeconds, false));
    }

    /*
     * Enables or disables plan of given index. If it's set to disabled no new stakes can be started with this plan,
     *  Cannot influence already ongoing stakes.
     *  Owner only.
     */
    function setPlanEnabled(uint256 planId, bool newPlanState) public onlyOwner {
        require(planExists(planId), "Plan does not exist");
        _plans[planId].enabled = newPlanState;
    }
}